Applying the New Trade Theories in Empirical CGE Models

Edward J. Balistreri

February 8, 2013

1 Calibration

Demand, output, and trade flows are observed:

$$P_{is}^0 Q_{is}^0 \equiv vafm_{is}$$

$$c_{ir}^0 Y_{ir}^0 \equiv vom_{ir}$$

$$\tilde{p}_{hrs}^0 \tilde{q}_{hrs}^0 N_{hrs}^0 \equiv [(1 + t_{hrs}^0) vxmd_{hrs}].$$

We also take as given the parameters a, b, and σ_h.

Assumed values of the f_{hrs} indicate benchmark firm-level revenues from the zero-cut-off profit condition:

$$\tilde{p}_{hrs}^0 \tilde{q}_{hrs}^0 = \frac{f_{hrs}(a + 1 - \sigma_h)}{a\sigma_h}, \tag{1}$$

and given the value of trade we can recover the number of firms

$$N_{hrs}^0 = \left[(1 + t_{hrs}^0) vxmd_{hrs}\right] \frac{a\sigma_h}{f_{hrs}(a + 1 - \sigma_h)}. \tag{2}$$

Notice that different assumptions about f_{hrs} simply rescale our measure of N_{hrs}^0.

The scale of M^0_{hr} is related to the flow of payments to sunk costs δf^s_{hr}. The free-entry condition establishes this relationship:

$$\delta f^s_{hr} = \sum_s \tilde{p}^0_{hrs} \tilde{q}^0_{hrs} \frac{N^0_{hrs}}{M^0_{hr}} \frac{\sigma_h - 1}{a\sigma_h}. \tag{3}$$

Given N^0_{hrs}, one can assume (measure) δf^s_{hr} to establish M^0_{hr} (being careful to ensure that $N^0_{hrs}/M^0_{hr} < 1 \ \forall \ \{h, r, s\}$),

... or one can assume (measure) one of the N^0_{hrs}/M^0_{hr} (which gives M^0_{hr}) to establish δf^s_{hr}.

We can use the ratio of operating to entered firms to calculate benchmark productivities,

$$\tilde{\varphi}^0_{hrs} = b \left(\frac{a}{a + 1 - \sigma_h} \right)^{1/(\sigma_h - 1)} \left(\frac{N^0_{hrs}}{M^0_{hr}} \right)^{-1/a}; \tag{4}$$

and this allows us to calculate the benchmark prices according to the optimal markup,

$$\tilde{p}^0_{hrs} = \frac{1 + t^0_{hrs}}{\tilde{\varphi}_{hrs}(1 - 1/\sigma_h)}. \tag{5}$$

The firm level quantity must be consistent with bilateral trade volumes;

$$\tilde{q}^0_{hrs} = \frac{(1 + t^0_{hrs}) vxmd_{hrs}}{\tilde{p}_{hrs} N^0_{hrs}}. \tag{6}$$

The only remaining calibration parameters are the λ_{hrs}, and these are recovered by inverting the demand functions;

$$\lambda_{hrs} = \frac{\tilde{q}^0_{hrs}(\tilde{p}^0_{hrs})^{\sigma_h}}{vafm_{hs}}, \tag{7}$$

where the empirical price index is elaborated to include these bilateral weights (rather than iceberg trade costs):

$$P_{hs} = \left[\sum_r \lambda_{hrs} N^0_{hrs} \tilde{p}^{1-\sigma_h}_{hrs}\right]^{1/(1-\sigma_h)}. \tag{8}$$

Exercises:

1. Verify that model.gms (from empirical.zip) runs and confirm that the total carbon leakage rates are 20% with no border adjustments and 14.5% with border adjustments.
2. Search for the string "*PE Calibration:" in model.gms to find the Melitz model calibration sequence.
   (a) Change the scale of M^0_{hr}. How does this affect the leakage rates?
   (b) What about doubling the assumed fixed costs, how does this affect the leakage rates?
   (c) Now try doubling just the assumed fixed costs associated with trade flows going into China (`CHN`).
3. Recalibrate the model assuming that $a(h) = 5$. How does this affect the leakage rates?

Table 1: Minimal Multiregion General Equilibrium with Alternative Trade Theories

Equation Description	Associated Variable	Equation Number: General	Armington	Krugman	Melitz	Dimensions
Unit Expenditure Function	U_r: Welfare	(27)				R
Final Demand	E_r: Consumer Price Index	(28)				R
Demand by Sector	P_{ir}: Good Price	(29)				$I \times R$
Composite Price Index	Q_{ir}: Aggregate Quantity		(4)	(6)	(12)	$I \times R$
Free Entry	N_{kr} or M_{hr}: Entered Firms			(9)	(24)	$(K + H) \times R$
Zero Cutoff Profits	N_{hrs}: Operating Firms				(21)	$H \times R \times R$
Firm-level Demand	p_{krs} or $\tilde{p}_{hrs}$: Firm Price			(7)	(13)	$(K + H) \times R \times R$
Firm-level Markup	q_{krs} or $\tilde{q}_{hrs}$: Firm Output			(8)	(14)	$(K + H) \times R \times R$
Firm-level Productivity	$\tilde{\varphi}_{hrs}$: Productivity				(18)	$H \times R \times R$
Composite-input Markets	c_{ir}: Unit-cost Index		(5)	(10)	(26)	$I \times R$
Unit-cost Function	Y_{ir}: Upstream Output	(30)				$I \times R$
Primary-factor Markets	w_{fr}: Factor price	(31)				$F \times R$
Income	GDP_r: Income	(32)				R

2 Computation:

There are excessive dimensions in empirical applications of new trade theory. Furthermore, there are non-convexities associated with productivity and variety effects. These conspire to make computation of the integrated general equilibrium difficult (if not impossible). A decomposition method can be employed, however, that separates the general equilibrium effects from the commodity-specific increasing-returns trade equilibrium.

Markusen and Rutherford's MPSGE *tricks*

```
$TITLE: Model M62-MPS: Large-Group Monopolistic Competition, uses MPS/GE
$ONTEXT
                        Production Sectors          Consumers
   Markets   |   XI     X        N        Y        W    |  CONS   ENTR
----------------------------------------------------------------------
        PX   |        100                        -100    |
        CX   !  100  -100                                |
        PY   |                          100      -100    |
        PF   |                  20                       |          -20
        PU   |                                    200    |  -200
        PW   |  -32              -8     -60              |   100
        PZ   |  -48             -12     -40              |   100
        MK   |  -20                                      |           20
$OFFTEXT
PARAMETERS
  ENDOW  Size index for the economy
  INDEX  Price index for the X goods
  EP     Elasticity of substitution among X varieties;
ENDOW = 1;
EP = 5;
$ONTEXT
$MODEL:M62
$SECTORS:
        X       ! Activity level for sector X
        Y       ! Activity level for sector Y
        W       ! Activity level for sector W (Hicksian welfare index)
        N       ! Activity level for sector X fixed costs, no. of firms
        XI      ! Activity level -- marginal cost of X
$COMMODITIES:
        PX      ! Price index for commodity X (gross of markup)
        CX      ! Marginal cost index for commodity X (net markup)
        PY      ! Price index for commodity Y
        PW      ! Price index for unskilled labor
        PZ      ! Price index for skilled labor
        PF      ! Unit price of inputs to fixed cost
        PU      ! Price index for welfare (expenditure function)21
$CONSUMERS:
        CONS    ! Income level for consumer CONS
        ENTRE   ! Entrepreneur (converts markup revenue to fixed cost)
$AUXILIARY:
        XQADJ   ! Quantity adjustment   (positive when X>1)
        XPADJ   ! X output subsidy rate (positive when X>1)
$PROD:X s:1
        O:PX    Q: 80   P:1.25   A:CONS  N:XPADJ  M:-1
        I:CX    Q: 80   P:1.25
$PROD:Y s:1
        O:PY    Q:100
        I:PW    Q: 60
        I:PZ    Q: 40
$PROD:XI s:1
```

```
        O:CX    Q: 80   A:ENTRE   T:0.20
        I:PW    Q: 32
        I:PZ    Q: 48
$PROD:N s:1
        O:PF    Q:20
        I:PZ    Q:12
        I:PW    Q: 8
$PROD:W s:1.0
        O:PU    Q:200
        I:PX    Q: 80   P:1.25
        I:PY    Q:100
$DEMAND:CONS
        D:PU    Q:200
        E:PW    Q:(100*ENDOW)
        E:PZ    Q:(100*ENDOW)
        E:PX    Q:80    R:XQADJ
$DEMAND: ENTRE
        D:PF    Q:20
$CONSTRAINT:XQADJ
        XQADJ =E= (N**(1/(EP-1)))*X - X;
$CONSTRAINT:XPADJ
        XPADJ =E= (N**(1/(EP-1))) - 1;
$OFFTEXT
$SYSINCLUDE mpsgeset M62
*       Adjust bounds so that the auxiliary variables can take on
*       negative values:
XQADJ.LO = -INF;
XPADJ.LO = -INF;
*       Benchmark replication:
PY.FX = 1;
PX.L = 1.25;
CX.L = 1.25;
$INCLUDE M62.GEN
SOLVE M62 USING MCP;
INDEX = (N.L*CX.L**(1-EP))**(1/(1-EP));
DISPLAY INDEX;
*       Counterfactual: expand the size of the economy
ENDOW = 2;
$INCLUDE M62.GEN
SOLVE M62 USING MCP;
INDEX = (N.L*CX.L**(1-EP))**(1/(1-EP));
DISPLAY INDEX;
```

Figure 1: A Decomposition Algorithm



Figure 2: Example of Convergence in Application



Exercises:

1. Calibrate a simple multiregion Armington model in MPSGE and run an experiment where you impose a tariff.
2. In the same program calibrate the A2.gms partial equilibrium Krugman equations to the same benchmark trade flows. (Be sure to add the tariff instrument).
3. Develop an iterative recalibration algorithm that allows us to find the full general equilibrium solution without incorporating the Krugman equations in the GE model (see the code below or model.gms for an example).
4. Formulate the full algebraic Krugman general equilibrium and confirm that the solution from the decomposition algorithm is accurate.

Tom's original decomposition example of the Melitz model:

```
$title Pilot Implementation of Bernard, Redding and Schott (Version 5)

$ontext

Reference: "Comparative Advantage and Heterogeneous Firms", Working
Paper, July 2004 by Andrew B. Bernard, Stephen Redding and Peter K.
Schott.

I found the model formulated in this paper to be very interesting, but
computationally infeasible. The problem is that the definition of an
equilibrium in a model with N regions and M markets involves NxNxM
entry decisions for monopolistic firms. The curse of dimensionality
suggests that while the results are intriguing, it may be impossible
to evaluate the model in an empirical context.


The file illustrates a decomposition method which provides some hope
for empirical applications of this framework.

I propose a decomposition through which we can solve multiregional,
multisectoral general equilibrium models with heterogeneous firms
engaging in monopolistic competitive markets. A partial equilibrium
model of firm-level, multi-regional monopolistic competition is
embedded within a general equilibrium model. The general equilibrium
model determines factor prices and comparative advantage. The partial
equilibrium model assesses productivity effects which arise from the
introduction of new varieties within a Dixit-Stiglitz demand
framework.

In the partial equilibrium model firm entry and exit decisions are
made on the basis of expected profitability, depending on productivity
levels drawn from a Pareto distribution. Exporting activities are
determined by firm-specific productivity levels. Only those firms
with sufficiently high productivity engage in trade, as trade flows
are subject to fixed costs.

The model incorporates Dixit-Stiglitz preferences in final demand with
firm-level product differentiation. Factor markets are competitive
and introduce conventional comparative advantage factors into the
model.

This version includes an upward sloping supply function for firm-level
resources in the partial equilibrium subproblem. The slope of the
supply function is taken as 5, although this parameter need not be
exact.

$offtext

set     i Industries /1,2/
        k Regions /H,F/
```

```
        f Primary factors /
                s Skilled labor,
                u Unskilled labor/,
        t(i,k,k) Set of active trade links,
        eq(i) Active industries;

eq(i) = yes;

alias (k,kk,kkk);

parameter

* Inputs describing Pareto distribution for firm productivity

*       g(phi) = (b/phi)**a * a/phi (36)
*
*       a > (sigma-1) so that log firm sales have a finite variance

* Standard deviation of the log of firm sales is 1/(a-sigma+1)

        a       Shape parameter in the Pareto distribution /3.4/
        b       Minimum productivity in the Pareto distribution /0.2/
        sigma   Intra-variety elasticity /3.8/
        beta(i,f) Factor intensities /1.s 0.6, 2.s 0.4/
        alpha(i) Value shares /1 0.5, 2 0.5/
        fe(i)   Fixed cost of entry /1 2, 2 2/
        fx      Fixed costs of export /0.2/
        delta   Probability of firm death /0.025/,
        eta     Elasticity of supply in PE submodel /5/
        rho     Inverse elasticity
        fc(i,k,k) Bilateral fixed cost,
        tau(i,k,k) Iceberg trade cost parameter;

fc(i,k,kk) = fx;
fc(i,k,k) = 0.1;

*       Begin with autarchy and then explore tau ranges from 2 to 1.2

*       tau(i,k,kk) = +inf;

tau(i,k,kk) = 2;
tau(i,k,k) = 1;
t(i,k,kk) = yes$(1/tau(i,k,kk));
beta(i,"u") = 1 - beta(i,"s");
rho = 1 - 1/sigma;

*       Parameters calibrated to match plant-level manufacturing data
*       in Bernard et al (2003)

table s(f,k) Primary factor endowments
        H       F
```

```
S       1200    1000
U       1000    1200;

VARIABLES
        P(i,k)          Price index
        PT(i,k,kk)      Weighted average price index
        QT(i,k,kk)      Mean bilateral sales
        M(i,k)          Total number of firms
        N(i,k,kk)       Number of firms
        THETA(i,k,kk)   Firm supply shares
        PHIS(i,k,kk)    Minimum productivity,
        PI(i,k,kk)      Bilateral profit
        PHIT(i,k,kk)    Weighted average productivity,
        Y(i,k)          Aggregate sectoral output,
        C(i,k)          Unit cost of firm inputs
        W(f,k)          Factor price
        R(k)            Regional income;

POSITIVE VARIABLES N, M, QT, W, P, C, Y, R;

EQUATIONS ptdef, pdef, qtdef, phisdef, pidef, phitdef, thetadef,
ndef, mdef, ydef, cdef, cdefpe, fmarket, income;

*       The unit cost of goods are defined on the basis of cost-minimizing
*       behavior with Cobb-Douglas technology:
cdef(i,k).. C(i,k) =e= prod(f, W(f,k)**beta(i,f));

*       Factor supply equals facgtor demand (the market clearance equations
*       are multiplied by the factor price to improve numerical stability):

fmarket(f,k).. W(f,k)*s(f,k) =e= sum(i, beta(i,f)*C(i,k)*Y(i,k));

*       Factor income is based on factor prices and endowments:

income(k).. R(k) =e= sum(f, W(f,k)*s(f,k));

*       Aggregate sectoral output is based on the number of firms. Sectoral
*       output consists of fixed costs of supply, fixed trade costs and
*       variable costs of suply which depend on the quantity shipped, the
*       trade costs and weighted average productity:

ydef(i,k)$eq(i)..
        Y(i,k) =e= M(i,k)* (delta*fe(i) + sum(t(i,k,kk),
                THETA(i,k,kk) * (fc(i,k,kk)+QT(i,k,kk)*tau(i,k,kk)/PHIT(i,k,kk))));

*       In the partial equilibrium model we fix industry output but assume that
*       supply responds to shadow value of output. This equation has an
*       exception operator eq(i) indicating that it only enters for the active
*       sector:

cdefpe(i,k)$eq(i)..
```

```
        Y.L(i,k)*(C(i,k)/C.L(i,k))**eta =e= M(i,k)* (delta*fe(i) +
          sum(t(i,k,kk), THETA(i,k,kk) * (fc(i,k,kk)+QT(i,k,kk)*tau(i,k,kk)/PHIT(i,k,kk))));

*       Free entry zero profit condition relates fixed cost of entry
*       to profits and -- this determines the aggregate number of firms
*       operating in region k:

mdef(i,k)$eq(i)..  C(i,k)*fe(i) =e= sum(t(i,k,kk), PI(i,k,kk) * THETA(i,k,kk))/delta;

*       Output per firm is constant in this model (due to the
*       assumption of iceberg trade cost and identical factor content
*       of fixed and variable costs), so the supply from kk to k
*       is the same as the number of firms:

pdef(i,k)$eq(i)..  sum(t(i,kk,k), N(i,kk,k)*PT(i,kk,k)**(1-sigma)) =e= P(i,k)**(1-sigma);

*       The cost of supply from k to kk depends on factor cost (C), the
*       transportation markup (tau), the weighted average productivity
*       (PHIT) and the markup factor (rho):

ptdef(t(i,k,kk))..      PT(i,k,kk) * (rho * PHIT(i,k,kk)) =e= tau(i,k,kk) * C(i,k);

*       The aggregate sales from k into kk follows from a nested
*       CES demand function in which the top-level preferences are
*       are Cobb-Douglas:
qtdef(t(i,k,kk))..     QT(i,k,kk) =e= (alpha(i)*R(kk)/P(i,kk)) * (P(i,kk)/PT(i,k,kk))**sigma;

*       Unit profits on individual firm sales from k to kk equal
*       markup revenue less fixed costs:

pidef(t(i,k,kk))..      PI(i,k,kk) =e= PT(i,k,kk)*QT(i,k,kk)/sigma - C(i,k)*fc(i,k,kk);

* The fraction of firms from market k selling into market kk:

thetadef(t(i,k,kk))..   THETA(i,k,kk) =e= (b/PHIS(i,k,kk))**a;

*       The number of firms selling from k to kk depend on the number of
*       firms operating in region k:

ndef(t(i,k,kk))..       N(i,k,kk) =e= THETA(i,k,kk) * M(i,k);

*       The minimum productivity level firm selling from region k
*       to region kk:

phisdef(t(i,k,kk))..    QT(i,k,kk)*PT(i,k,kk) * ((a+1-sigma)/a) =e= sigma*fc(i,k,kk)*C(i,k);

*       The weighted average productivity is computed on the basis of
*       the cutoff productivity level for firms in the industry.

phitdef(t(i,k,kk))..    PHIT(i,k,kk) =e= (a/(a+1-sigma))**(1/(sigma-1)) * PHIS(i,k,kk);
```

```
*       Define two models, one which includes the full systems and a second
*       which defines a partial equilibrium model for a single commodity:

MODEL   MCHF /  pdef.P, ndef.N, ptdef.PT, qtdef.QT, pidef.PI, thetadef.THETA,
                phisdef.PHIS, phitdef.PHIT, ydef.Y,
                mdef.M, cdef.C, fmarket.W, income.R /,

        MCHFPE /pdef.P, ndef.N, ptdef.PT, qtdef.QT, pidef.PI, thetadef.THETA,
                phisdef.PHIS, phitdef.phit, mdef.M, cdefpe.C/;

*       Finally, specify a general equilibrium relaxation and verify that
*       it is calibrated when inialized at the solution of the integrated model:

PARAMETER
        aref(i,k) Reference output for Armington sector
        dref(i,k,kk) Reference demand
        pdref(i,k) Reference price;

*       The following is an MPSGE model in which we evaluate factor
*       price equilibria given industry structure in each of the
*       individual industries.

$ontext
$model:MCHFGE
$sectors:
        D(i,k) ! Aggregate demand (CES)
        Y(i,k) ! Industry supply

$commodities:
        P(i,k) ! Composite price
        C(i,k) ! Output price
        W(f,k) ! Factor wages

$consumers:
        R(k) ! Representative agent

*       Agents are endowed with factors and demand goods. This
*       demand function is exogenously specified and unchanged
*       through the iterative procedure:

$demand:R(k) s:1
        e:W(f,k)        q:s(f,k)
        d:P(i,k)        q:alpha(i)

*       Producers combine primary factor inputs to produce
*       commodity outputs. These cost functions are exogenous
*       and unchanged in the iterative sequence.

$prod:Y(i,k) S:1
        o:C(i,k)        Q:1
        i:W(f,k)        q:beta(i,f) p:1
```

```
*       Iterative adjustments of the GE model occur within
*       the "implicit Armington aggregator".

*       Domestic demand substitutes across goods from
*       different firms. AREF, DREF and PDREF are based
*       on current equilibrium values from the industry-
*       level partial equilibrium models. Increases in
*       the total number of firms supplying market k
*       are reflected in the value of aref(). Changes in
*       the sourcing of inputs from different regions are
*       represented by the reference price-quantity pairs.
*       The underlying model is based on firm-level
*       product differentiation, so while this model appears
*       to have the structure of a conventional Armington
*       model, entry and exit decisions will endogenously
*       alter the implied structure of the Armington model.

$prod:D(i,k) s:sigma
        o:P(i,k)        q:aref(i,k)
        i:C(i,kk)       q:dref(i,kk,k) p:pdref(i,kk)
$offtext
$sysinclude mpsgeset MCHFGE

*       We now verify that we can compute equilibria using
*       block Gauss-Seidel decomposition. These calculations
*       involve partial equilibrium models (one for each commodity)
*       followed by a general equilibrium calculation which assesses
*       the comparative advantage effects in a seemingly conventional
*       CRTS Armington trade model.
*       In this program we first compute the equilibrium with
*       this decomposition procedure, and we then evaluate precision
*       of the estimation in an integrated model.

PT.L(i,k,kk) = 1;
QT.L(i,k,kk) = 1;
N.L(i,k,kk) = 1;
PHIS.L(i,k,kk) = 1;
PHIT.L(i,k,kk) = 1;
R.L(k) = sum(f,s(f,k));
Y.L(i,k) = alpha(i) * R.L(k);
M.L(i,k) = Y.L(i,k);
C.LO(i,k) = 0.00001;
W.LO(f,k) = 0.00001;
P.LO(i,k) = 0.00001;
PT.LO(i,k,kk) = 0.00001;
PHIS.LO(i,k,kk) = b;
PHIT.LO(i,k,kk) = b;

*       Initialize productivity levels using autarky values:
```

```
PHIS.L(i,k,kk) = 0;
PHIT.L(i,k,kk) = 0;
PHIS.L(i,k,kk)$t(i,k,kk) = b * (fc(i,k,kk)/(fe(i)*delta) * (sigma-1)/(a+1-sigma))**(1/a);
PHIT.L(i,k,kk)$t(i,k,kk) = (a/(a+1-sigma))**(1/(sigma-1)) * PHIS.l(i,k,kk);
THETA.L(i,k,kk)$t(i,k,kk) = 1;

*       The partial equilibrium relaxation has fixed income
*       levels and fixed numbers of firms:

alias (i,ii);

*       Partial equilibrium solution holds marginal cost of commodities, income
*       and aggregate production of each commodity fixed:

R.FX(k) = R.L(k);
Y.L(i,k) = alpha(i) * R.L(k)/C.L(i,k);

set iter Outer iterations /iter0*iter5/;

parameter       iterlog         Iteration log
                nfirm           Iteration log (number of firms);

loop(iter,

*       Produce an iteration log:

        iterlog(iter,"R"," ",k) = R.L(k);
        iterlog(iter,"Y",i,k) = Y.L(i,k);
        iterlog(iter,"M",i,k) = M.L(i,k);
        iterlog(iter,"C",i,k) = C.L(i,k);

*       One subproblem for each commodity:
        loop(ii,
          eq(i) = no;
          eq(ii) = yes;
          t(i,k,kk) = no;
          t(ii,k,kk) = yes$(1/tau(ii,k,kk));
          SOLVE MCHFPE using MCP;
        );

        nfirm(iter,i) = M.L(i,"h");

*       Recalibrate the general equilibrium model to the
*       partial equilibrium solution:

        aref(i,k) = alpha(i)*R.L(k)/P.L(i,k);
        dref(i,k,kk) = PT.L(i,k,kk)*QT.L(i,k,kk)*N.L(i,k,kk)/C.L(i,k);
        pdref(i,k) = C.L(i,k);

*       Remove bounds to compute the general equilibrium solution:
```

```
        Y.l(i,k) = sum(kk, dref(i,k,kk));
        R.UP(k) = +inf; R.LO(k) = 0;
        R.FX(k)$(ord(k)=1) = R.L(k);

$include MCHFGE.GEN
        SOLVE MCHFGE USING MCP;

*       Fix prices and activity levels which enter the partial equilibrium
*       model as exogenous parameters:

        t(i,k,kk) = yes$(1/tau(i,k,kk));
        R.FX(k) = R.L(k);
);

*       Check this solution using the fully-specified model:

eq(i) = yes;
t(i,k,kk) = yes$(1/tau(i,k,kk));
R.UP(k) = +inf; R.LO(k) = 0;
MCHF.ITERLIM = 0;
SOLVE MCHF USING MCP;

*       Generate a plot:

$libinclude plot nfirm

*       Display iterative sequence:

option iterlog:1:1:3;
display iterlog;
```